# $Id: PKGBUILD 90049 2013-05-06 19:37:57Z foutrelis $
# Maintainer:  Bartłomiej Piotrowski
# Contributor: Stefan Husmann <stefan-husmann@t-online.de>

pkgname=jbig2dec
pkgver=0.11
pkgrel=6
pkgdesc="decoder implementation of the JBIG2 image compression format"
arch=('x86_64' 'i686')
url="http://jbig2dec.sourceforge.net/"
license=('GPL3')
depends=('glibc' 'libpng')
options=('!libtool')
source=(https://github.com/rillian/jbig2dec/archive/$pkgver.tar.gz
        $pkgname-$pkgver-libpng15.patch)
sha256sums=('8af7b94b6026aff718d4b41ad990f2702f8fd4a348d565c4b48a695988ca0e9a'
            'ed871abd9419dbf35ba7f38d4835098d3838fd9400856ac4ec029e63bb42d859')
prepare() {
  cd $pkgname-$pkgver
  patch -Np0 -i "$srcdir"/$pkgname-$pkgver-libpng15.patch
  sed -i 's/AM_CONFIG_HEADER/AC_CONFIG_HEADER/' configure.ac
}

build() {
  cd $pkgname-$pkgver
  ./autogen.sh --prefix=/usr
  make
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR="$pkgdir/" install
}
